Pergamon

0098-3004(94)E0013-J

Computers & Geosciences Vol. 20, No. 6, pp. 905–917, 1994

0098-3004/94 $7.00 + 0.00

COMPUTATION OF LONGSHORE CURRENTS AND SEDIMENT TRANSPORT

FERNANDO J. CAVIGLIA*

Comisión de Investigaciones Científicas de la Provincia de Buenos Aires y Facultad de Ciencias Naturales y Museo, UNLP, Casilla de Correo 45 (1900) La Plata, Argentina

(Received 26 July 1993; accepted 28 December 1993)

Abstract—A numerical model for the steady-state profile of the longshore current induced by regular, obliquely incident, breaking waves is presented. The wave parameters must be given at an arbitrary depth. A rapid convergent numerical algorithm is described for the solution of the governing equation. The model is solved using a nonlinear bottom friction law in which the friction coefficients are a function of the bottom roughness which is computed at each point using an empirical formula. The predicted current profiles are combined with some of the known formulae for sediment transport computations.

Key Words: Longshore currents, Sediment transport, Numerical model, Forward sweep implicit technique.

INTRODUCTION

If breaking waves approach a straight coastline at an oblique angle, they induce a longshore current in the surf zone. The current acts somewhat analogous to a river, transporting sediment mobilized by the breaking waves, and is confined to a zone with a width of two to three times the width of the surf zone. The prediction of such longshore currents and the associated sediment transport is of prime importance for coastal engineers.

The purpose of this paper is to present a numerical model of longshore current in combination with some of the known formulae for predicting sediment transport which can be used for engineering purposes. The model is solved on a desktop computer and requires only data available in any engineering project.

LONGSHORE CURRENT MODEL

According to the generally employed assumptions in longshore current modeling, namely (i) longshore uniformity in waves and bottom topography, (ii) negligible bottom friction in the cross-shore direction, and (iii) applicability of linear-wave theory, the vertically integrated, time-averaged momentum equations for nearshore water motion (Mei, 1983) become

$$\rho g d \frac{\mathrm{d}\eta}{\mathrm{d}x} = -\frac{\mathrm{d}S_{xx}}{\mathrm{d}x}, \tag{1}$$

$$\frac{\mathrm{d}}{\mathrm{d}x}\left[A_{\mathrm{h}} d \frac{\mathrm{d}V}{\mathrm{d}x}\right] - \frac{\tau_{\mathrm{b}y}}{\rho} = \frac{1}{\rho}\frac{\mathrm{d}S_{xy}}{\mathrm{d}x}, \tag{2}$$

for the cross-shore x component and for the longshore y component respectively (see Fig. 1), where ρ is the water density, g the acceleration resulting from gravity, $d = h + \eta$ the mean water depth, h the still water depth, η the elevation of the free surface above the still water level, A_{h} the eddy viscosity coefficient, V the longshore current velocity and $\tau_{\mathrm{b}y}$ the bottom frictional stress. The radiation stress components S_{xx} and S_{xy} are given by

$$S_{xx} = E[n(\cos^2\alpha + 1) - \tfrac{1}{2}], \tag{3}$$

$$S_{xy} = \frac{\sin\alpha}{C} F_x, \tag{4}$$

where $F_x = -ECn\cos\alpha$ is the wave energy flux towards the shoreline, C the phase velocity of waves, $n = \frac{1}{2} + (kd/Sh2kd)$, α the angle that the wave front makes with the shoreline, $E = \frac{1}{8}\rho g H^2$ the energy density of the waves per unit area, H the wave height, $k = 2\pi/L$ the wave number and L the wavelength. The phase velocity is calculated with the approximated expression introduced by Visser (1984)

$$C = \begin{cases} \sqrt{gd}\left(1 - \dfrac{d}{L_0}\right) & \text{as } \dfrac{d}{L_0} \leqslant 0.36 \\ \dfrac{gT}{2\pi} = C_0 & \text{as } \dfrac{d}{L_0} > 0.36 \end{cases}, \tag{5}$$

where T is the wave period and the subscript 0 denotes a value on deep water.

Wave set-up

Equation (1) governs the mean water surface displacement η. According to the specific assumptions of this paper namely, (iv) constant beach slope, (v)

*Address for correspondence: Laboratorio de Oceangrafia Costera y Estuarios, Casilla de Correo 45 (1900) La Plata, Argentina.

proportionality between the wave height and the mean water depth in the surf zone, $H = \gamma d$, (vi) negligible wave set-down offshore of the plunge line, x_p, and (vii) applicability of the shallow water approximation, $n = 1$, $\cos\alpha \cong 1$ and $C = \sqrt{gd}$, the equation for the mean water surface displacement can be integrated to obtain

$$\eta = Ks(x_p - x), \tag{6}$$

where $K = [1 + 8/3\gamma^2]^{-1}$, γ the breaker index and s the beach slope. If Equation (6) is evaluated at $x = x_s$, the following relation can be obtained

$$x_s = Kx_p, \tag{7}$$

where x_s is the line of maximum wave set-up. The plunge line is calculated with the following expression (Svendsen, 1987; Larson and Kraus, 1989)

$$x_p = x_b - 3H_b, \tag{8}$$

where H_b is the breaking wave height, $x_b = (d_b/s + x_s)$



Figure 1. Definition sketch for coordinate system and nearshore region, where η is wave set-up, h still-water depth, x_b width of breaker zone, x_s distance from shoreline to still-water line, x_p distance from shoreline to plunge line, and α angle of incidence of waves.

the breaker zone width and d_b the depth at breaking ($d_b = H_b/\gamma$).

Substitution of (8) into (7), taking into account the expression for x_b, gives

$$x_s = \frac{3\gamma^2}{8}\left[\frac{d_b}{s} - 3H_b\right]. \tag{9}$$

From (6) and (7), the mean water depth can be written as

$$\bar{d} = \begin{cases} \left(1 - \frac{\bar{x}_s}{P}\right) s\bar{x}\frac{x_b}{d_b} & \text{as} \quad \bar{x} \leqslant P \\ (\bar{x} - \bar{x}_s)s\frac{x_b}{d_b} & \text{as} \quad \bar{x} > P \end{cases}, \tag{10}$$

where the following dimensionless variables were introduced

$$\bar{d} = \frac{d}{d_b}, \quad \bar{x} = \frac{x}{x_b}, \quad P = \frac{x_p}{x_b}, \quad \bar{x}_s = \frac{x_s}{x_b}. \tag{11}$$

Driving force

The longshore driving force resulting from obliquely incident wave is

$$\frac{\partial S_{xy}}{\partial x} = \frac{\sin \alpha_i}{C_i}\frac{\partial F_x}{\partial x}, \tag{12}$$

where the subscript i denotes a value at an arbitrary depth.

Substituting the expressions of the wave energy flux towards the shoreline and the energy density of the waves per unit area, in combination with the assumption (v) and the shallow water approximation, into (12) gives

$$\frac{\partial S_{xy}}{\partial x} = \frac{\sin \alpha_i}{C_i}\begin{cases} -\frac{5}{16}\rho g^{3/2}\gamma^2\left[\frac{d_b}{x_b}\right]^{5/2} x^{3/2} & \text{as} \quad x \leqslant x_b \\ 0 & \text{as} \quad x > x_b \end{cases}. \tag{13}$$

Visser (1984) introduced the following assumptions: (viii) the dissipation of wave energy takes place shoreward of the plunge line, (ix) dF_x/dx is proportional to $x^{3/2}$ shoreward of the plunge line and vanishes offshore of this line, and (x) the transport of wave energy towards the shore is given by the transport of wave energy predicted by linear wave theory in the cross-shore x-direction. Under these assumptions follows from Equation (13) that

$$\frac{\partial S_{xy}}{\partial x} = \frac{\sin \alpha_i}{C_i}\begin{cases} -D_b\left[\frac{\bar{x}}{P}\right]^{3/2} & \text{as} \quad \bar{x} \leqslant P \\ 0 & \text{as} \quad \bar{x} > P \end{cases}, \tag{14}$$

where

$$D_b = \frac{5}{2}\frac{1}{Px_b}E_i C_i n_i \cos \alpha_i.$$

Bottom friction term

Under the combined action of waves and currents the time-averaged bottom friction in longshore direction can be expressed as

$$\begin{aligned} \frac{\tau_{by}}{\rho} = \frac{1}{8}F\frac{1}{T}\int_0^T & [V^2 + 2\xi u_m V \sin \alpha \cos \omega t \\ & + (\xi u_m)^2 \cos^2 \omega t]^{1/2} \\ & * (V + \xi u_m \sin \alpha \cos \omega t)\,dt, \end{aligned} \tag{15}$$

where $\omega = 2\pi/T$ is the angular frequency, $u_m = \pi/T\ H/Shkd$ the maximum orbital velocity near the bottom and $\xi = 2(F_w/F)^{1/2}$ with F_w being the Jonsson friction coefficient and F the Darcy–Weisbach friction coefficient.

Because the elliptic integral (15) must be solved frequently in the numerical longshore current program in this paper the square wave approximation introduced by Nishimura (1988) is used. Then, the time-averaged bottom friction in longshore direction becomes

$$\frac{\tau_{by}}{\rho} = \frac{1}{8}F\left(Z + \frac{z^2 \sin^2 \alpha}{Z}\right)V, \tag{16}$$

$$Z = \tfrac{1}{2}[(V^2 + z^2 + 2zV \sin \alpha)^{1/2} + (V^2 + z^2 - 2zV \sin \alpha)^{1/2}].$$

and $z = (2/\pi)\xi u_m$.

The most popular formulae for predicting the friction coefficients are

$$F_w = \begin{cases} \exp\left[-5.977 + 5.213\left(\frac{r}{a_b}\right)^{0.194}\right] & \text{as} \quad \frac{r}{a_b} \leqslant 0.63 \\ 0.3 & \text{as} \quad \frac{r}{a_b} > 0.63 \end{cases} \tag{17}$$

$$F = 8\left[2.5 \ln \frac{12d}{r}\right]^{-2}, \tag{18}$$

where $a_b = u_m/\omega$ is the horizontal water particle excursion at the bottom and r the equivalent diameter of the roughness elements. Equation (17) was suggested by Jonsson (1980) and Equation (18) is based on Nikuradse's experiments. According to Nielsen (1985) the total roughness, r, is calculated from the following expression

$$r = 190D\sqrt{\theta' - 0.05} + 8\frac{\eta_r^2}{\lambda_r}, \tag{19}$$

where η_r is the ripple height, λ_r the ripple length, D the mean sediment grain diameter, $\theta' = \frac{1}{2}F'_w\Psi$ the so-called skin friction Shields parameter, F'_w the Jonsson friction coefficient based on $r = 2.5D$, $\Psi = u_m^2/[(s-1)gD]$ the mobility parameter and s the relative sediment density ($s = \rho_s/\rho$, where ρ_s is the sediment density). If $\theta' \leqslant 0.05$ then $r = 2.5D$, which is the roughness for a flatbed of fixed sand grains. Nielsen (1981) recommended the following formulae for prediction of ripple geometry

$$\frac{\eta_r}{\lambda_r} = 0.342 - 0.34\sqrt[4]{\theta'}, \tag{20}$$

$$\frac{\eta_r}{a_b} = \begin{cases} 0.275 - 0.022\sqrt{\Psi} & \text{as} \quad \Psi \leq 10 \\ 21\Psi^{-1.85} & \text{as} \quad \Psi \geq 10 \end{cases} \tag{21}$$

This author recommended to base the ripple calculations on the significant wave height for field situations.

The proposed expressions for ripple geometry calculation are valid for nonbreaking wave conditions. Ripples are washed out when the mobility parameter, Ψ, is larger than about 200–250 (see Dingler and Inman, 1976). Van Rijn (1989) assumed that the ripple existence is limited to values of Ψ smaller than 250. Then, from Equation (19)

$$r = 190D\sqrt{\theta' - 0.05}, \tag{22}$$

which is the formula recommended by Nielsen (1985) for the prediction of the roughness of flat, mobile sand beds. Under breaking wave conditions the mobility parameter, in general, will be larger than 250. Furthermore, it is assumed that the longshore current has no influence upon the ripple geometry.

Lateral mixing

The first term of Equation (2) is an empirical closure relation describing lateral mixing resulting from Reynolds stresses. Visser (1984) expressed the eddy viscosity coefficient as

$$A_h = \mathrm{M}qd, \tag{23}$$

where M is a constant of order 3 and q the characteristic turbulent velocity. The nondimensional form of A_h is

$$\bar{A}_h = \frac{A_h}{A_0}, \tag{24}$$

where A_0 can be defined in arbitrary form.

The eddy viscosity coefficient resulting from this parameterization decreases rapidly offshore of the plunge line. This result seems to be in agreement with the laboratory measurements of Deguchi, Sawaragi, and Ono (1992), who determined that A_h becomes maximum near $d/d_b = 0.7$–0.8 and decreases rapidly toward offshore.

NUMERICAL SOLUTION

Substitution of (14), (16), and (24) into (2) yields

$$\frac{d^2\bar{V}}{d\bar{x}^2} + g_1(\bar{x})\frac{d\bar{V}}{d\bar{x}} - g_2(\bar{x})\bar{V} = \begin{cases} -g_3(\bar{x}) & \text{as} \quad \bar{x} \leq P \\ 0 & \text{as} \quad \bar{x} > P \end{cases}, \tag{25}$$

in which $\bar{V} = V/V_0$,

$$g_1(\bar{x}) = \frac{1}{\bar{A}_h\bar{d}}\frac{d\bar{A}_h\bar{d}}{d\bar{x}}, \tag{26}$$

$$g_2(\bar{x}) = C_1\frac{F\left(Z + \dfrac{z^2\sin^2\alpha}{Z}\right)}{\bar{A}_h\bar{d}}, \tag{27}$$

$$g_3(\bar{x}) = \frac{1}{\bar{A}_h\bar{d}}\left[\frac{\bar{x}}{P}\right]^{3/2}, \tag{28}$$

with

$$C_1 = \frac{1}{8}\frac{x_b^2}{A_0 d_b} \quad \text{and} \quad V_0 = \frac{D_b x_b^2}{\rho A_0 d_b}\frac{\sin\alpha_i}{C_i}.$$

The differential equation to be solved is nonlinear because $g_2(\bar{x})$ contains V through Z. The differential equation is solved as a linear equation. This is done by solving (25) and (27) with an iteration procedure. The canonical form of the differential equation for (25) for calculation of the grid cell number is

$$g_{4_i}\bar{V}_{i-1} + g_{5_i}\bar{V}_i + g_{6_i}\bar{V}_{i+1} = g_{7_i}, \tag{29}$$

$$g_{4_i} = 1 - \tfrac{1}{2}g_{1_i}\Delta\bar{x}, \quad g_{5_i} = -(2 + g_{2_i}\Delta\bar{x}^2)$$

$$g_{6_i} = 1 + \tfrac{1}{2}g_{1_i}\Delta\bar{x}$$

and

$$g_{7_i} = \begin{cases} -g_{3_i}\Delta\bar{x}^2 & \text{as} \quad \bar{x} \leq P \\ 0 & \text{as} \quad \bar{x} > P \end{cases}$$

Equation (29) is solved for each calculation cell, with i extending from 2 to $N - 1$, for a grid encompassing N cells. Boundary conditions for $\bar{V}$ must be provided at cells 1 and N. The boundary conditions are

$$\bar{V} = 0 \quad \text{as} \quad \bar{x} = 0 \quad \text{or} \quad i = 1$$

$$\bar{V} = 0 \quad \text{as} \quad \bar{x} \to \infty \quad \text{or} \quad i \to \infty$$

with $\bar{x} = (i - 1)\Delta\bar{x}$.

Because for $\bar{x} > 2$ the longshore current velocity $\bar{V}$ becomes small (Visser, 1984) it is assumed that

$$\bar{V} = 0 \quad \text{as} \quad \bar{x} = 3 \quad \text{or} \quad i = 241$$

if $\Delta\bar{x} = 1/80$.

For $i = 2$–240, Equation (29) is a tridiagonal set of linear equations in the unknowns $\mathbb{V} = [\bar{V}_2, \bar{V}_3, \ldots, \bar{V}_{N-1}]$ and can be written in matrix form as $\mathbb{M}\mathbb{V} = \mathbb{G}$ where $\mathbb{G} = [g_{7_2}, g_{7_3}, \ldots, g_{7_{N-1}}]$. The matrix $\mathbb{M}$ is tridiagonal with principal diagonal elements g_{5_i}, the elements to left of the diagonal are g_{4_i} and to the right are g_{6_i}. All the other elements are zero. This system is solved easily by forward sweep and a backward substitution. The algorithm is simply Gaussian elimination. The backward substitution solves for $\bar{V}$, and it is given by

$$\bar{V}_i = U_i - W_i\bar{V}_{i+1} \quad \text{for} \quad i = 2\text{–}240, \tag{30}$$

where

$$W_i = \frac{g_{6i}}{g_{5_i} - W_{i-1}g_{4_i}} \quad \text{and} \quad U_i = \frac{g_{7_i} - U_{i-1}g_{4_i}}{g_{5_i} - W_{i-1}g_{4_i}},$$

with

$$\bar{V}_{241} = W_1 = U_1 = 0.$$

Table 1. Limiting wave height (from Van Rijn, 1989)

	Breaker index		
Wave steepness	s = 1/100	s = 1/20	s = 1/10
$H_{s0}/L_0 = 0.002$	0.60	0.70	0.80
0.004	0.60	0.70	0.80
0.006	0.60	0.70	0.80
0.010	0.60	0.70	0.80
0.020	0.50	0.65	0.80
0.040	0.40	0.55	0.70
0.060	0.35	0.50	0.70

The forward sweep reduces $\mathbb{M}$ to an upper-triangular matrix with principal diagonal elements unity and elements to the right of the diagonal given by W_i. The current velocity calculated with (30), previously multiplied by V_0, is replaced into (27) and the procedure described here is repeated again. This procedure is carried out until the absolute value of the difference between $\bar{V}_i$ and the velocity calculated in the preceding iteration step be less than $\epsilon = 0.01$. In general, the solution converges after no more than eight iterations.

FIELD SITUATIONS

A question which arises in comparing monochromatic wave models with field random waves is the determination of an appropriate wave height parameter for model calculations. Wu, Thornton, and Guza (1985) gave a detailed explanation for selecting the root-mean square wave height, $H_{r.m.s.}$, as the appropriate one.

Another related problem is the appropriate breaker index for random waves. Thornton, Wu, and Guza (1984) presented a criteria for random breaking waves based on field and laboratory data. From their results, Van Rijn (1989) derived the values of $\gamma_s (\gamma_s = H_{bs}/d_b$, where H_{bs} is the significant breaking wave height) shown in Table 1. Multiple linear regression gives

$$\gamma_s = 0.603 + 2.147s + (30.685s - 5.192)\left[\frac{H_{0s}}{L_0}\right]. \tag{31}$$

where the correlation coefficient is 0.99.

For a Rayleigh wave height probability distribution, the relationship between significant wave height and root-mean square wave height is $H_s \cong \sqrt{2}H_{r.m.s.}$. Thornton, Wu, and Guza (1984) showed that this relation could be used with an error of about $\pm 5\%$. From this relation, it is assumed that

$$\gamma_{r.m.s.} = \gamma_s/\sqrt{2}, \tag{32}$$

where $\gamma_{r.m.s.}$ is the breaker index based on $H_{r.m.s.}$.

When the input conditions are those at the breaker line a simple predictive formula is used to estimate the deep water wave height prior to determining the breaker index (Larson and Kraus, 1989)

$$H_{bs} = 0.53H_{0s}\left[\frac{H_{0s}}{L_0}\right]^{-0.24}. \tag{33}$$

USING THE MODEL

The program has been written in Microsoft QuickBASIC 4.50. The user is requested to input the following parameters: the significant wave height (H_s in meters) or the root-mean square wave height ($H_{r.m.s.}$ in meters), the angle that the wave front makes with the shoreline (α in degrees), the wave period (T in seconds), the depth in which wave measurements were made (meters) if the wave conditions are not at the breaker line, the beach slope, the mean sediment diameter (D in millimeters), the particle diameter D_{90} (10% by weight exceeded in size, millimeters) which is not strictly necessary for the program to be run, and the sediment density (ρ_s in kilogram per cubic meter). With these parameters the program computes the longshore current and sediment transport distributions, the mean longshore current in m/s and the longshore sediment transport rate in m^3/s and m^3/day.

The program includes a routine which allows a screen view of the longshore current and sediment transport distributions. This routine requires a VGA or MCGA display adapter.

As an example of the program capability a print screen of the resulting longshore current distribution for the Thornton and Guza (1986, 1989) 4 February data are shown in Figure 2. Wave conditions at 9.1-m depth were: $H_{r.m.s.} = 0.52$ m, $T = 14.2$ sec and $\alpha = 18.4°$, with s = 0.038, $D = 0.23$ mm (Gable, 1981) and $\rho_s = 2650$ kg/m^3. The data are superimposed on the graph.

DISCUSSION

The specialized problem of studying longshore current generation in the surf zone is solved by the use of a forward sweep implicit technique. The program presented here is short, fast, easy to use, and versatile. The numerical model can be adapted easily



Figure 2. Print screen of computed longshore current distribution (——) for Thornton and Guza (1986, 1989) February 4 data (■).

to other eddy viscosity coefficients, wave energy dissipation models, bottom shear stress theories, etc.

The present model for longshore current computations was combined with any of the known formulae for predicting sediment transport distribution. Some of them are described briefly in the Appendix.

REFERENCES

Bijker, E. W., 1967, Some considerations about scales for coastal models with movable bed: Publ. No. 50, Delft Hydraulics Laboratory, Delft, The Netherlands, 147 p.

Deguchi, I., Sawaragi, T., and Ono, M., 1992, Longshore current and lateral mixing in the surf zone (abst.): Proc. 23rd Coastal Engrg. Conf., A.S.C.E. (Venice, Italy), Book of Abstracts, Paper No. 172, p. 353–354.

Dingler, J. R., and Inman, B. L., 1976, Wave-formated ripples in nearshore sands: Proc. 15th Coastal Engrg. Conf., A.S.C.E. (Honolulu, Hawaii), p. 2109–2126.

Gable, C. G., ed., 1981, Report on data from the Nearshore Sediment Transport Study experiment on Leadbetter Beach, Santa Barbara, California, January–February 1980: IMR Ref. 80-5, Univ. California, Inst. Marine Resources, La Jolla, California, 314 p.

Jonsson, I. G., 1980, A new approach to oscillatory, rough turbulent boundary layers: Ocean Engrg., v. 7, no. 1, p. 109–152.

Larson, M., and Kraus, N. C., 1989, SBEACH: numerical model for simulating storm-induced beach change: Rept. 1, U.S. Army Coastal Eng. Res. Center, Washington, D.C. Tech. Rept. CERC-89-9, 255 p.

Mei, C. C., 1983, The applied dynamics of ocean surface waves: John Wiley & Sons, Inc. New York, Ch. 10, p. 451–503.

Nielsen, P., 1981, Dynamics and geometry of wave generated ripples: Jour. Geophys. Res., v. 86, no. C7, p. 6467–6472.

Nielsen, P., 1985, A short manual of coastal bottom boundary layers and sediment transport: Public Works Dept., Tech. Memo. 85/1, N.S.W., 56 p.

Nishimura, H., 1988, Computation of nearshore current, *in* Horikawa, K., ed., Nearshore dynamics and coastal processes: Univ. Tokyo Press, Tokyo, Japan, p. 271–291.

Svendsen, I. A., 1987, Analysis of surf zone turbulence: Jour. Geophys. Res., v. 92, no. C5, p. 5115–5124.

Swart, D. H., 1976, Predictive equations regarding coastal transport: Proc. Coastal Engrg. Conf., A.S.C.E. (Honolulu, Hawaii), p. 1113–1132.

Thornton, E. B., and Guza, R. T., 1986, Surf zone longshore currents and random waves: Field data and models: Jour. Phys. Oceanography, v. 16, no. 7, p. 1165–1178.

Thornton, E. B., and Guza, R. T., 1989, Models for surf zone dynamics, *in* Seymour, R. J., ed., Nearshore sediment transport: Plenum Press, New York, p. 337–369.

Thornton, E. B., Wu, C. S., and Guza, R. T., 1984, Breaking design criteria: Proc. 19th Coastal Engrg. Conf., A.S.C.E. (Houston, Texas), p. 31–41.

U.S. Army Corps of Engineers, 1984, Shore Protection Manual: Wtrwy. Expr. Station, Vicksburg, Mississippi, v. 1, Ch. 4, p. 89–99.

Van Rijn, L. C., 1989, Sediment transport by currents and waves: Handbook, Rept. H 461, Delft Hydraulics, Delft, The Netherlands, Ch. 2, p. 15–19 and Ch. 6, p. 15–21.

Visser, P. J., 1984, A mathematical model of uniform longshore currents and the comparison with laboratory data: Rept. No. 84-2, Dept. Civ. Engrg., Delft Univ. Tech., Delft, The Netherlands, 151 p.

Watanabe, A., 1992, Total rate and cross-shore distribution of longshore sand transport (abst.): Proc. 23rd Coastal Engrg. Conf., A.S.C.E. (Venice, Italy), Book of Abstracts, Paper No. 297, p. 619–620.

Wu, C. S., Thornton, E. B., and Guza, R. T., 1985, Waves and longshore currents: Comparison of a numerical model with field data: Jour. Geophys. Res., v. 90, no. C3, p. 4951–4958.

APPENDIX

Sediment Transport Formulae

Because details on the applied formulae are given in the original papers, the formulae are described only briefly in this Appendix.

Adapted Engelund–Hansen Formula

The formula for the total sediment transport q_T of the adapted Engelund–Hansen method is

$$q_T = V\frac{0.05C_h\tau_r^2}{\rho^2 g^{5/2}\Delta^2 D}, \tag{A1}$$

where

$$\tau_r = \frac{1}{8}\rho F V^2\left[1 + \frac{1}{2}\left(\xi\frac{u_m}{V}\right)^2\right],$$

$$C_h^2 = \frac{8g}{F} \quad \text{and } \Delta = s - 1.$$

Bijker Formula

Bijker (1967) suggested to use the Kalinske–Frijlink formula for bedload computation

$$q_b = 5D\frac{V}{C_h}g^{1/2}\exp\left[\frac{-0.27\Delta D\rho g}{\mu\tau_r}\right], \tag{A2}$$

where $\mu = (C_h/C_{90})^{1.5}$, $C_{90} = 18\log(12d)/D_{90}$. Bijker (1967) used the Einstein expression for suspended load. Swart (1976) proposed a modification of the procedure for calculations of suspended load. He used the following expression

$$c = c_a\left(\frac{z}{a}\right)^{-b}, \tag{A3}$$

where c is the sediment concentration at height z, c_a the concentration at a reference height a and

$$b = 1.05Z^{0.96}\left[\frac{a}{d}\right]^{0.013Z}, \tag{A4}$$

with $Z = w_s/0.4V_{*,r}$, $V_{*,r} = (\tau_r/\rho)^{1/2}$ and w_s the sediment fall velocity.

The bedload is assumed to take place in a layer with thickness $a = r$ above the bed. From Bijker (1967)

$$c_a = \frac{q_b}{6.35V_* r}, \tag{A5}$$

so the suspended load becomes

$$q_s = \frac{q_b}{6.35(1-b)}\left[\left(\frac{d}{r}\right)^{1-b} - 1\right]\frac{V}{V_*}, \tag{A6}$$

where $V_* = [1/8\,F]^{1/2}V$. The total load is $q_T = q_b + q_s$.

Watanabe Formula

The sediment transport formula proposed by Watanabe (1992) is

$$q_T = A_c\left[\frac{(\tau - \tau_{cr})V}{\rho g}\right], \tag{A7}$$

where A_c is a dimensionless coefficient of order 2, τ the maximum value of bottom friction under the combined action of waves and longshore current with the roughness length equal to the grain diameter, $\tau_{cr} = (\rho_s - \rho)gD\theta_{cr}$ the critical shear stress for the onset of general sand movement and θ_{cr} the Shields parameter. Here τ_r is used instead of τ.

CERC Formula

The CERC formula is an integral method which relates the total sediment transport across the surf zone with the longshore component of the wave energy flux at the breaker line (U.S. Army Corps of Engineers, 1984)

$$Q_T = K \frac{P_{lb}}{(\rho_x - \rho)ga'}, \tag{A8}$$

where K is a dimensionless constant, $a' = 0.65$ the sand porosity and $P_{lb} = E_b C_b \sin\alpha_b \cos\alpha_b$ the longshore component of the wave energy flux. The subscript b indicates that all evaluations are to be made at the breaker line. A value of $K = 0.77$ must be used if the computation of P_{lb} is based on the $H_{r.m.s.}$ wave height. For reducing the wave height to the breaker line is used the usual breaking criterion for spilling breaking waves, $\gamma = 0.78$.

Program Listing

```
'     A QuickBASIC 4.50 program to compute longshore currents
'     and the associated sediment transport

'     S    -Beach slope (non-dimensional);
'     HO1 -Wave height to an arbitrary depth (meters);
'     HB   -Wave height at the breaker line (meters);
'     TA   -Angle that wave front makes with the shore line (degree);
'     T    -Wave period (seconds);
'     DO1 -Depth of the wave measurements (meters);
'     D50 -Mean sediment diameter (millimeters);
'     D90 -Particle diameter such that 10% by weight exceeded
'          in size (millimeters);
'     RSED-Sediment density (kilogram per cubic meter).
'     V(I)-Nondimensional current velocity

    CLS
    LOCATE 12, 15: INPUT "Do you want to see the instructions-(Y or N)"; P$
    IF P$ = "N" OR P$ = "n" GOTO 5

    CLS : LOCATE 4, 25: PRINT "Input parameters:"
    LOCATE 8, 10
    PRINT "A-Significant wave height or root-mean square wave height (m)."
    LOCATE 9, 10
    PRINT "B-Angle that wave front makes with the shore line (degree)."
    LOCATE 10, 10: PRINT "C-Wave period (s)."
    LOCATE 11, 10: PRINT "D-Beach slope."
    LOCATE 12, 10: PRINT "E-Mean sediment diameter, D50 (mm)."
    LOCATE 13, 10
    PRINT "F-Particle diameter such that 10% by weight exceeded"
    LOCATE 14, 10:
    PRINT "  in size, D90 (mm). It is not strictly necessary for the"
    LOCATE 15, 10: PRINT "  program to be run."
    LOCATE 16, 10: PRINT "G-Sediment density (kg/m3)."
    LOCATE 17, 10
    PRINT "H-Depth of wave measurements (m). It is necessary only when the"
    LOCATE 18, 10: PRINT "  input conditions are not at the breaker line."
    LOCATE 25, 10: INPUT "Press ENTER to continue"; P$

5   DIM H(241), ALF(241), FC(241), FW(241), UM(241), D(241), AH(241)
    DIM W(241), U(241), V(241), V1(241), QB(241), QW(241), QH(241), R(241)

10  CLS : LOCATE 8, 15: PRINT "Wave height parameter"
    LOCATE 11, 10: PRINT "Significant wave height ........(1)"
    LOCATE 12, 10: PRINT "Root-mean square wave height ...(2)"
    LOCATE 13, 10: INPUT "                                  "; D$: CLS
    LOCATE 12, 10: INPUT "Are conditions at the breaker line-(Y or N)"; C$

    CLS : LOCATE 12, 15: INPUT "Beach slope"; S
    CLS : LOCATE 12, 15: INPUT "Wave height"; HO1
    CLS : LOCATE 12, 15: INPUT "Wave angle"; TA
    CLS : LOCATE 12, 15: INPUT "Wave period"; T
    CLS : LOCATE 12, 15: INPUT "Mean sediment diameter"; D50
    CLS : LOCATE 12, 12
    INPUT "Is the sediment diameter D90 available-(Y or N)"; P1$
    CLS : IF P1$ = "Y" OR P1$ = "y" THEN LOCATE 12, 20: INPUT "D90"; D90
    CLS : LOCATE 12, 15: INPUT "Sediment density"; RSED
    IF C$ = "Y" OR C$ = "y" GOTO 15
    CLS : LOCATE 12, 15: INPUT "Depth of wave measurements"; DO1
```

```
15  IF D$ = "1" THEN HO1 = HO1 / (2 ^ .5)
    D50 = D50 / 1000: D90 = D90 / 1000
    TO1 = TA * ATN(1) / 45
    IF C$ = "Y" OR C$ = "y" THEN HB = HO1: TB = TO1
    LO = 9.8 * (T ^ 2) / (8 * ATN(1)): CO = LO / T
    AA1 = .603 + 2.147 * S
    BB1 = 30.685 * S - 5.192

    IF C$ = "N" OR C$ = "n" GOTO 20
    HO = ((2 ^ .5) * HB * (LO ^ -.24) / .53) ^ (1 / .76)
    MU = AA1 + BB1 * (HO / LO): MU = MU / (2 ^ .5): HO = HO / (2 ^ .5)
    PB = HB / MU
    CB = ((9.8 * PB) ^ .5) * (1 - PB / LO)
    GOTO 30

'       Deep water wave condition computation

20  IF DO1 >= LO / 2 THEN TO = TO1: HO = HO1: GOTO 25
    CO1 = ((9.8 * DO1) ^ .5) * (1 - DO1 / LO)
    IF DO1 / LO > .36 THEN CO1 = CO
    LO1 = CO1 * T
    ALF5 = SIN(TO1) * CO / CO1
    TO = ATN(ALF5 / ((1 - ALF5 ^ 2) ^ .5))
    KK = 8 * ATN(1) * DO1 / LO1
    SH1 = 2.7182818# ^ (2 * KK): SH2 = 2.7182818# ^ -(2 * KK)
    SH = (SH1 - SH2) / 2: GX = 1 + 2 * KK / SH: CGO1 = .5 * CO1 * GX
    HO = HO1 * ((2 * CGO1 * COS(TO1) / (CO * COS(TO))) ^ .5)

'       Breaking wave computation

25  MU = AA1 / (2 ^ .5) + BB1 * (HO / LO)
    HB = HO

   DO
    HB1 = HB
    PB = HB / MU
    CB = ((9.8 * PB) ^ .5) * (1 - PB / LO)
    ALF7 = SIN(TO) * CB / CO
    TB = ATN(ALF7 / ((1 - ALF7 ^ 2) ^ .5))
    HB = HO * ((CO * COS(TO) / (2 * CB * COS(TB))) ^ .5)
   LOOP WHILE ABS(HB1 - HB) > .001

30  LP = 3 * HB
    XS = 3 * (MU ^ 2) * (PB / S - 3 * HB) / 8
    XB = PB / S + XS
    XP = XB - LP
    XS = XS / XB: P = XP / XB

'       Plunge line search

    DX = 1 / 80
     J = 0
     PP = P
   DO
     PP = PP - DX: J = J + 1
   LOOP UNTIL PP <= DX
     P = J * DX
     IP = J + 1
     J = 0: PP = 0

     CGI = CGO1: TI = TO1: CI = CO1: HI = HO1: EI = 9.8 * (HO1 ^ 2) / 8
     EB = 9.8 * (HB ^ 2) / 8: EO = 9.8 * (HO ^ 2) / 8
     IF C$ = "Y" OR C$ = "y" THEN EI = EB: TI = TB: CGI = CB
     IF C$ = "Y" OR C$ = "y" THEN HI = HB: CI = CB
     IF HO1 = HO THEN EI = EO: TI = TO: CGI = CO / 2: CI = CO: HI = HO
     DB = 2.5 * EI * CGI * COS(TI) / (P * XB)
     M = 3
     B1 = 3 / (M * (((1 - XS / P) * S) ^ 2)): B2 = 3 / ((S ^ 2) * M)
     R1 = -.5 + (.25 + B1) ^ .5: R2 = -.5 - (.25 + B2) ^ .5
     A12 = R2 - R1 * (1 - XS / P)
     A1 = ((1 - XS / P) * 1.5 - R2) * (P ^ -R1) / A12
     A2 = ((P - XS) ^ (1 - R2)) * ((1.5 - R1) / P) / A12
```

```
     AHB = CB * PB
     DELTA = (RSED - 1025) / 1025

    FOR I = 2 TO 240
     X = (I - 1) * DX

'       Non-dimensional mean water depth computation

     IF X <= P THEN D(I) = (1 - XS / P) * S * X * XB / PB: GOTO 35
     D(I) = (X - XS) * S * XB / PB

'       Phase velocity computation

35   C = ((9.8 * PB * D(I)) ^ .5) * (1 - D(I) * PB / LO)
     IF D(I) * PB / LO > .36 THEN C = CO

'       Incidence's angle computation

     ALF = SIN(TI) * C / CI
     ALF(I) = ATN(ALF / ((1 - ALF ^ 2) ^ .5))

'       Wave height computation

     LO11 = C * T: KK = 8 * ATN(1) * PB * D(I) / LO11
     IF I <= 81 THEN H(I) = HB * D(I): GOTO 40
     SH1 = 2.7182818# ^ (2 * KK): SH2 = 2.7182818# ^ -(2 * KK)
     SH = .5 * (SH1 - SH2): GX = 1 + 2 * KK / SH: CG = .5 * C * GX
     H(I) = HO1 * ((CGI * COS(TI) / (CG * COS(ALF(I)))) ^ .5)

'       Maximun orbital velocity computation at the bottom

40   SH1 = 2.7182818# ^ KK: SH2 = 2.7182818# ^ -KK
     SH = .5 * (SH1 - SH2)
     UM(I) = 4 * ATN(1) * H(I) / (T * SH)

'       Non-dimensional eddy viscosity coefficient computation

     IF I > IP THEN QCIN = (A2 * ((X - XS) ^ R2)) ^ (1 / 3): GOTO 45
     QCIN = (A1 * (X ^ R1) + (X / P) ^ 1.5) ^ (1 / 3)
45   AH(I) = M * QCIN * D(I) * PB * (DB ^ (1 / 3)) / AHB

'       Ripple and roughness computations

     UMB = UM(I) * (2 ^ .5)
     AB = UMB * T / (8 * ATN(1))
     F1 = (UMB ^ 2) / (DELTA * 9.8 * D50)
     FW1 = 2.7182818# ^ (5.213 * ((2.5 * D50 / AB) ^ .194) - 5.977)
     T1 = .5 * F1 * FW1
     IF F1 >= 250 THEN ET = 0: GOTO 50
     IF F1 < 10 THEN ET = AB * (.275 - .022 * (F1 ^ .5)): GOTO 50
     ET = 21 * AB * (F1 ^ -1.85)
50   EL = .342 - .34 * (T1 ^ .25)
     IF T1 < .05 THEN R(I) = 2.5 * D50: GOTO 55
     R(I) = 8 * EL * ET + 190 * D50 * ((T1 - .05) ^ .5)

'       Friction coefficients computations

55   FC(I) = 8 * ((2.5 * LOG(12 * PB * D(I) / R(I))) ^ -2)
     ABC = R(I) / AB
     IF ABC > .63 THEN FW(I) = .3: GOTO 60
     FW(I) = 2.7182818# ^ (-5.977 + 5.213 * (ABC ^ .194))

60  NEXT I

     VO = DB * (XB ^ 2) * SIN(TI) / (AHB * PB * CI)
     C1 = .125 * (XB ^ 2) / (AHB * PB)
     V(1) = 0
     W(1) = 0: U(1) = V(1)

   DO

'       Computation of the elements of the tridiagonal and
'       upper-triangular matrices
```

```
    J = 0
   FOR I = 2 TO 240
    X = (I - 1) * DX
    EP = 2 * ((FW(I) / FC(I)) ^ .5)
    FI = EP * UM(I)
    Z1X = FI / (2 * ATN(1))
    Z1 = (VO * V(I)) ^ 2 + 2 * Z1X * SIN(ALF(I)) * VO * V(I) + Z1X ^ 2
    Z2 = (VO * V(I)) ^ 2 - 2 * Z1X * SIN(ALF(I)) * VO * V(I) + Z1X ^ 2
    Z = .5 * (Z1 ^ .5 + Z2 ^ .5)
    FT = Z + ((Z1X * SIN(ALF(I))) ^ 2) / Z
    G2 = C1 * FC(I) * FT / (AH(I) * D(I))
    G11 = AH(I + 1) * D(I + 1) - AH(I - 1) * D(I - 1)
    G12 = 2 * DX * AH(I) * D(I)
    G1 = G11 / G12
    IF I > IP THEN G3 = 0: GOTO 65
    G3 = ((X / P) ^ 1.5) / (AH(I) * D(I))
65  G4 = 1 - .5 * G1 * DX
    G5 = -(2 + G2 * (DX ^ 2))
    G6 = 1 + .5 * G1 * DX
    G7 = -G3 * (DX ^ 2)
    W(I) = G6 / (G5 - G4 * W(I - 1))
    U(I) = (G7 - G4 * U(I - 1)) / (G5 - G4 * W(I - 1))
   NEXT I

'      Backward substitution for computing the non-dimensional
'      longshore current velocity

    V(241) = 0
   FOR I = 240 TO 2 STEP -1
    V(I) = U(I) - W(I) * V(I + 1)
    IF ABS(V(I) - V1(I)) > .01 THEN J = J + 1
    V1(I) = V(I)
   NEXT I

  LOOP WHILE J <> 0

    I = 0: MM = 0
70 DO
    MM = MM + V(I) * VO
    I = I + 1
    IF I <= 81 GOTO 70
  LOOP WHILE (V(I) * VO) > .01
    FIN = I - 1
    VM = MM / FIN

'      Fall sediment velocity computation

    VISC = .0000014
    B = 9.8 * DELTA * (D50 ^ 3) / (VISC ^ 2)
    IF B < 39 THEN WS = 9.8 * DELTA * (D50 ^ 2) / (18 * VISC): GOTO 75
    IF B > 10000 THEN WS = (9.8 * DELTA * D50 / .91) ^ .5: GOTO 75
    WS = ((9.8 * DELTA) ^ .7) * (D50 ^ 1.1) / (6 * (VISC ^ .4))

75  BB = 0: HH = 0: WW = 0

'      Computation of the critical shear stress for the onset
'      of sediment movement

    DAST = D50 * ((9.8 * DELTA / (VISC ^ 2)) ^ (1 / 3))
    IF DAST <= 4 THEN TICR = .24 / DAST: GOTO 80
    IF DAST <= 10 THEN TICR = .14 * (DAST ^ -.64): GOTO 80
    IF DAST <= 20 THEN TICR = .04 * (DAST ^ -.1): GOTO 80
    IF DAST <= 150 THEN TICR = .013 * (DAST ^ .29): GOTO 80
    TICR = .055
80  TBCR = DELTA * 9.8 * D50 * TICR

'      Sediment transport computation

   FOR I = 2 TO FIN
    EP = 2 * ((FW(I) / FC(I)) ^ .5)
    TC = .125 * FC(I) * ((V(I) * VO) ^ 2)
    TR = TC * (1 + .5 * ((EP * UM(I) / (V(I) * VO)) ^ 2))
    VR = TR ^ .5: VC = TC ^ .5
```

```
'       Bijker formula
     CHE = (8 * 9.8 / FC(I)) ^ .5
     IF P1$ = "N" OR P1$ = "n" GOTO 85
     IF TR <= TBCR GOTO 85
     B11 = WS / (.4 * VR)
     B1 = 1.05 * (B11 ^ .96) * ((R(I) / (D(I) * PB)) ^ (.013 * B11))
     B12 = 1 - B1
     Z1 = ((D(I) * PB / R(I)) ^ B12) - 1
     FC90 = 8 * ((2.5 * LOG(12 * D(I) * PB / D90)) ^ -2)
     CH90 = (8 * 9.8 / FC90) ^ .5
     NU = (CHE / CH90) ^ 1.5
     QBB = 5 * D50 * VC
     QB = QBB * (2.7182818# ^ -(.27 * 9.8 * DELTA * D50 / (NU * TR)))
     QS = QB * Z1 * V(I) * VO / (6.35 * B12 * VC)
     QB(I) = QB + QS
     IF QB(I) > QB(I - 1) THEN QBMAX = QB(I)
85   BB = BB + QB(I)

'       Watanabe formula

     IF TR <= TBCR GOTO 90
     QT = 2 * (TR - TBCR) * V(I) * VO / 9.8
     QW(I) = QT
     IF QW(I) > QW(I - 1) THEN WAMAX = QW(I)
90   WW = WW + QW(I)

'       Adapted Engelund-Hansen formula

     QH1 = ((9.8 * DELTA) ^ 2) * D50 * (9.8 ^ .5)
     QH(I) = .05 * CHE * (TR ^ 2) * V(I) * VO / QH1
     IF QH(I) > QH(I - 1) THEN QEMAX = QH(I)
     HH = HH + QH(I)

    NEXT I

'       CERC Formula

     CB1 = (9.8 * HB / .78) ^ .5
     IF C$ = "Y" OR C$ = "y" THEN EB1 = EB: TB1 = TB: GOTO 95
     HB1 = (((.78 / 9.8) ^ .5) * (HI ^ 2) * CGI * COS(TI)) ^ .4
     CB1 = (9.8 * HB1 / .78) ^ .5
     EB1 = .125 * 9.8 * (HB1 ^ 2)
     ALF8 = SIN(TI) * CB1 / CI
     TB1 = ATN(ALF8 / ((1 - ALF8 ^ 2) ^ .5))
95   PLB = EB1 * CB1 * SIN(TB1) * COS(TB1)
     CERC = .77 * PLB / (DELTA * 9.8 * .65)

     QH = HH * DX * XB
     QW = WW * DX * XB
     QB = BB * DX * XB

     CLS
     QBY = INT(QB * 86400): QHY = INT(QH * 86400)
     QWY = INT(QW * 86400): CERCY = INT(CERC * 86400)
     LOCATE 5, 5
     PRINT "Root-mean square wave height at the breaker line ="
     LOCATE 5, 55: PRINT USING "##.##"; HB
     LOCATE 6, 5: PRINT "Angle of incidence of wave at the breaker line ="
     LOCATE 6, 54: PRINT USING "##.#"; TB * 45 / ATN(1)
     LOCATE 7, 5: PRINT "Period of wave = "
     LOCATE 7, 22: PRINT USING "##.#"; T
     LOCATE 8, 5: PRINT "Breaker zone width ="
     LOCATE 8, 26: PRINT USING "###.#"; XB
     LOCATE 9, 5: PRINT "Breaker index ="
     LOCATE 9, 21: PRINT USING "#.##"; MU
     LOCATE 10, 5: PRINT "Mean longshore current velocity ="
     LOCATE 10, 39: PRINT USING "#.##"; VM
     LOCATE 12, 15: PRINT "Sediment transport"
     LOCATE 13, 14: PRINT "(m3/seg and m3/day)"
     IF P1$ = "N" OR P1$ = "n" GOTO 100
     LOCATE 15, 5: PRINT "Bijker formula                  ="
     LOCATE 15, 40: PRINT QB: LOCATE 15, 55: PRINT QBY
```

```
100  LOCATE 16, 5: PRINT "Adapted Engelund-Hansen formula ="
     LOCATE 16, 40: PRINT QH: LOCATE 16, 55: PRINT QHY
     LOCATE 17, 5: PRINT "Watanabe formula                ="
     LOCATE 17, 40: PRINT QW: LOCATE 17, 55: PRINT QWY
     LOCATE 18, 5: PRINT "CERC formula                    ="
     LOCATE 18, 40: PRINT CERC: LOCATE 18, 55: PRINT CERCY

     LOCATE 25, 5: INPUT "Paper output-(Y or N)"; YY$
     TB = TB * 45 / ATN(1)
     IF YY$ = "N" OR YY$ = "n" GOTO 110

     LPRINT TAB(10); "Breaking wave height = "; USING "#.##"; HB
     LPRINT TAB(10); "Angle of incidence at breaking = "; USING "##.#"; TB
     LPRINT TAB(10); "Period of wave = "; USING "##.#"; T
     LPRINT TAB(10); "Breaker zone width = "; USING "###.#"; XB
     LPRINT TAB(10); "Breaker index = "; USING "#.##"; MU
     LPRINT TAB(10); "Mean longshore current velocity = "; USING "#.##"; VM
     LPRINT
     LPRINT TAB(30); "Sediment transport"
     LPRINT TAB(29); "(m3/seg and m3/day)"
     LPRINT
     IF P1$ = "N" OR P1$ = "n" GOTO 105
     LPRINT TAB(10); "Bijker formula="; TAB(45); QB; TAB(60); QBY
105  LPRINT TAB(10); "Engelund-Hansen formula="; TAB(45); QH; TAB(60); QHY
     LPRINT TAB(10); "Watanabe formula="; TAB(45); QW; TAB(60); QWY
     LPRINT TAB(10); "CERC formula="; TAB(45); CERC; TAB(60); CERCY

110  CLS
     LOCATE 12, 15: PRINT "Do you want to see the longshore current"
     LOCATE 13, 15
     INPUT "and sediment transport distributions-(Y or N)"; P$
     IF P$ = "N" OR P$ = "n" GOTO 150

     SCREEN 11
115  CLS : LOCATE 12, 8: PRINT "Longshore current distribution.......(1)"
     LOCATE 13, 8: PRINT "Sediment transport distribution......(2)"
     LOCATE 14, 44: INPUT ; A2
     CLS : IF A2 = 1 GOTO 125

     IF P1$ = "N" OR P1$ = "n" GOTO 120
     LOCATE 10, 15: PRINT "Bijker formula..................(1)"
120  LOCATE 11, 15: PRINT "Watanabe formula................(2)"
     LOCATE 12, 15: PRINT "Adapted Engelund-Hansen formula..(3)"
     LOCATE 13, 47: INPUT ; A3

125  CLS
     LINE (100, 350)-(500, 350): LINE (100, 350)-(100, 360)
     LINE (500, 350)-(500, 360): LINE (300, 350)-(300, 360)
     LINE (200, 350)-(200, 355): LINE (400, 350)-(400, 350)

     LOCATE 24, 12.5: PRINT "0.0": LOCATE 24, 62: PRINT "2.0"
     LOCATE 24, 37.25: PRINT "1.0": LOCATE 25.4, 36.5: PRINT "X/ Xb"
     IF A2 = 1 THEN X$ = "V": Y$ = "---": Z$ = "Vo"
     IF A2 = 1 THEN LOCATE 9, 53: PRINT "Vo = "; USING "#.##"; VO
     IF A2 = 1 THEN LOCATE 9, 65: PRINT "m/s": GOTO 130

     LOCATE 4, 20: PRINT "Sediment transport distribution"
     X$ = "q": Y$ = "---": Z$ = "qo"

     IF A3 = 1 THEN Q01 = QBMAX
     IF A3 = 2 THEN Q01 = WAMAX
     IF A3 = 3 THEN Q01 = QEMAX

     LOCATE 9, 53: PRINT "qo = "; USING "#.#####"; Q01
     LOCATE 9, 65: PRINT "m": LOCATE 8.5, 65.5: PRINT 2
     LOCATE 9, 67: PRINT "/s"

130  IF A2 = 1 THEN LOCATE 4, 20: PRINT "Longshore current distribution"
     IF A2 = 1 GOTO 135

     IF A3 = 1 THEN LOCATE 5, 27: PRINT "Bijker formula"
     IF A3 = 2 THEN LOCATE 5, 26: PRINT "Watanabe formula"
     IF A3 = 3 THEN LOCATE 5, 20: PRINT "Adapted Engelund-Hansen formula"
```

```
135  LOCATE 10, 53: PRINT "Xb ="; USING "###.#"; XB
     LOCATE 10, 65: PRINT "m"
     LOCATE 14.5, 4: PRINT X$: LOCATE 14.501, 3: PRINT Y$
     LOCATE 15.5, 3.2: PRINT Z$

     LINE (100, 350)-(100, 100): LINE (100, 350)-(90, 350)
     LINE (100, 100)-(90, 100): LINE (100, 225)-(90, 225)
     LINE (100, 287.5)-(95, 287.5): LINE (100, 162.5)-(95, 162.5)

     LOCATE 22.51, 8.5: PRINT "0.0": LOCATE 6.51, 8.5: PRINT "1.0"
     LOCATE 14.51, 8.5: PRINT "0.5"

     LOCATE 11, 53: PRINT "Hb ="; USING "##.##"; HB
     LOCATE 11, 65: PRINT "m"
     LOCATE 12, 53: PRINT "αb = "; USING "##.#"; TB
     LOCATE 12, 65: PRINT "°"
     LOCATE 13, 53: PRINT "T  = "; USING "##.#"; T
     LOCATE 13, 65: PRINT "s"
     LOCATE 14, 53: PRINT "S  = 1:"; USING "###.#"; 1 / S

     IF A2 = 1 THEN LINE (405, 111)-(548, 111): LINE (405, 111)-(405, 235)
     IF A2 = 1 THEN LINE (548, 111)-(548, 235)
     IF A2 = 1 THEN LINE (405, 235)-(548, 235): GOTO 140
     LINE (405, 108)-(560, 108): LINE (405, 108)-(405, 235)
     LINE (560, 108)-(560, 235): LINE (405, 235)-(560, 235)

140  V1(1) = 350

    FOR I = 2 TO FIN
     IF A2 = 1 THEN A1 = V(I)
     IF A3 = 1 THEN A1 = QB(I) / QBMAX
     IF A3 = 2 THEN A1 = QW(I) / WAMAX
     IF A3 = 3 THEN A1 = QH(I) / QEMAX
     IF I > 161 GOTO 145
     X = 100 + 200 * (I - 1) / 80
     X1 = 100 + 200 * (I - 2) / 80
     V1(I) = 350 - 250 * A1
     LINE (X1, V1(I - 1))-(X, V1(I))
    NEXT I

145  LOCATE 30, 10: INPUT "More graphics-(Y or N)"; P$
     IF P$ = "Y" OR P$ = "y" THEN A2 = 0: A3 = 0: GOTO 115

150  A2 = 0: A3 = 0
     CLS : SCREEN 0
     LOCATE 12, 15: INPUT "More calculations-(Y or N)"; XX$
     IF XX$ = "Y" OR XX$ = "y" THEN CLEAR: GOTO 10
     CLS : END
```